
Description

These functions compute derivatives of the real function f (p) at the real parameter values p.

deriv init() begins the definition of a problem and returns D, a problem-description handle set
that contains default values.

The deriv init *(D, . . . ) functions then allow you to modify those defaults. You use these
functions to describe your particular problem: to set the identity of function f (), to set parameter
values, and the like.

deriv(D, todo) then computes derivatives depending upon the value of todo.

deriv(D, 0) returns the function value without computing derivatives.

deriv(D, 1) returns the first derivatives, also known as the gradient vector for scalar-valued
functions (type d and v) or the Jacobian matrix for vector-valued functions (type t).

deriv(D, 2) returns the matrix of second derivatives, also known as the Hessian matrix;
the gradient vector is also computed. This syntax is not allowed for type t evaluators.

The deriv result *(D) functions can then be used to access other values associated with the
solution.

Usually you would stop there. In other cases, you could compute derivatives at other parameter values:

deriv_init_params(D, p_alt)
deriv(D, todo)

Aside: The deriv init *(D, . . . ) functions have two modes of operation. Each has an optional
argument that you specify to set the value and that you omit to query the value. For instance, the
full syntax of deriv init params() is

void deriv_init_params(D, real rowvector parameters)
real rowvector deriv_init_params(D)

The first syntax sets the parameter values and returns nothing. The second syntax returns the previously
set (or default, if not set) parameter values.

All the deriv init *(D, . . . ) functions work the same way.

2 deriv( ) — Numerical derivatives

Syntax

D = deriv init()

(varies) deriv init evaluator(D
[
, &function()

]
)

(varies) deriv init evaluatortype(D
[
, evaluatortype

]
)

(varies) deriv init params(D
[
, real rowvector parameters

]
)

(varies) deriv init argument(D, real scalar k
[
, X

]
)

(varies) deriv init narguments(D
[
, real scalar K

]
)

(varies) deriv init weights(D
[
, real colvector weights

]
)

(varies) deriv init h(D
[
, real rowvector h

]
)

(varies) deriv init scale(D
[
, real matrix scale

]
)

(varies) deriv init bounds(D
[
, real rowvector minmax

]
)

(varies) deriv init search(D
[
, search

]
)

(varies) deriv init usemin(D
[
, {"off" | "on"}

]
)

(varies) deriv init min(D
[
, real rowvector min

]
)

(varies) deriv init verbose(D
[
, {"on" | "off"}

]
)

(varies) deriv init technique(D
[
, technique

]
)

(varies) deriv(D, {0 | 1 | 2})
real scalar deriv(D, {0 | 1 | 2})

real scalar deriv result value(D)

real vector deriv result values(D)

void deriv result values(D, v)

real rowvector deriv result gradient(D)

void deriv result gradient(D, g)

real matrix deriv result scores(D)

void deriv result scores(D, S)

real matrix deriv result Jacobian(D)

void deriv result Jacobian(D, J)

real matrix deriv result Hessian(D)

void deriv result Hessian(D, H)

real rowvector deriv result h(D)

real matrix deriv result scale(D)

real matrix deriv result delta(D)

real scalar deriv result errorcode(D)

string scalar deriv result errortext(D)

real scalar deriv result returncode(D)

void deriv query(D)

where D, if it is declared, should be declared

transmorphic D

and where evaluatortype optionally specified in deriv init evaluatortype() is

evaluatortype Description

"d" function() returns scalar value
"v" function() returns colvector value
"t" function() returns rowvector value

The default is "d" if not set.

and where search optionally specified in deriv init search() is

search Description

"interpolate" use linear and quadratic interpolation to search
for an optimal delta

"bracket" use a bracketed quadratic formula to search for
an optimal delta

"off" do not search for an optimal delta

The default is "interpolate" if not set.

and where technique optionally specified in deriv init technique() is

technique Description

"numdiff" numerical differentiation method
"complex" complex step method

The default is "numdiff" if not set.

Remarks and examples stata.com

Remarks are presented under the following headings:

First example
Notation and formulas

Notation
Numerical differentiation method
Complex step method

Type d evaluators
Example of a type d evaluator
Type v evaluators
User-defined arguments
Example of a type v evaluator
Type t evaluators
Example of a type t evaluator
Example of using step-size lower bounds
Example of complex step method
Functions

deriv init( )
deriv init evaluator( ) and deriv init evaluatortype( )
deriv init argument( ) and deriv init narguments( )
deriv init weights( )
deriv init params( )
Advanced init functions

deriv init h( ), . . . scale( ), . . . bounds( ), and . . . search( )
deriv init usemin( ) and deriv init min( )
deriv init verbose( )
deriv init technique( )

deriv( )
deriv( )

deriv result value( )
deriv result values( ) and deriv result values( )
deriv result gradient( ) and deriv result gradient( )
deriv result scores( ) and deriv result scores( )
deriv result Jacobian( ) and deriv result Jacobian( )
deriv result Hessian( ) and deriv result Hessian( )
deriv result h( ), . . . scale( ), and . . . delta( )
deriv result errorcode( ), . . . errortext( ), and . . . returncode( )
deriv query( )

First example

The derivative functions may be used interactively.

Below we use the functions to compute f ′(x) at x = 0, where the function is

f (x) = exp(−x2 + x− 3)

: void myeval(x, y)
> {
> y = exp(-x^2 + x - 3)
> }

: D = deriv_init()

: deriv_init_evaluator(D, &myeval())

: deriv_init_params(D, 0)

: dydx = deriv(D, 1)

http://stata.com

: dydx
.0497870683

: exp(-3)
.0497870684

The derivative, given the above function, is f ′(x) = (−2×x+1)×exp(−x2+x−3), so f ′(0) = exp(−3).

Notation and formulas

Notation

We wrote the above in the way that mathematicians think, that is, differentiate y = f (x). Statisticians,
on the other hand, think differentiate s = f (b). To avoid favoritism, we will write v = f (p) and write
the general problem with the following notation:

Differentiate v = f (p) with respect to p, where

v: a scalar
p: 1× np

The gradient vector is g = f ′(p) = df/dp, where

g: 1× np

and the Hessian matrix is H = f ′′(p) = d2f/dpdp′, where

H: np× np

deriv() can also work with vector-valued functions. Here is the notation for vector-valued functions:

Differentiate v = f (p) with respect to p, where

v: 1× nv, a vector
p: 1× np

The Jacobian matrix is J = f ′(p) = df/dp, where

J: nv× np

and where

J[ i, j ] = dv[ i ]/dp[ j ]

Second-order derivatives are not computed by deriv() when used with vector-valued
functions.

Numerical differentiation method

The classical method to compute numerical derivatives is the numerical differentiation method, which
is our default setting for technique. This section considers numerical derivatives using this method.

deriv() uses the following formula for computing the numerical derivative of f () at p

f ′(p) =
f (p + d)− f (p− d)

2d

where we refer to d as the delta used for computing numerical derivatives. To search for an optimal
delta, we decompose d into two parts.

d = h× scale

By default, h is a fixed value that depends on the parameter value.

h = (abs(p)+1e-3)*1e-3

deriv() searches for a value of scale that will result in an optimal numerical derivative, that is, one
where d is as small as possible subject to the constraint that f (x + d)− f (x− d) will be calculated to
at least half the accuracy of a double-precision number. This is accomplished by searching for scale
such that | f (x)− f (x− d)| falls between v0 and v1, where

v0 = (abs(f(x))+1e-8)*1e-8
v1 = (abs(f(x))+1e-7)*1e-7

Use deriv init h() to change the default h values. Use deriv init scale() to change the
default initial scale values. Use deriv init bounds() to change the default bounds (1e-8, 1e-7)
used for determining the optimal scale.

Sometimes, the computed delta d used for calculating numerical derivatives may be too small,
especially in the denominator. This may cause unstable computation results. You can avoid this by
bounding d with a minimum value. Use deriv init usemin() to set whether to use minimum
values for d. When minimum values are requested, the default minimums are 1e-6 for first-order
derivatives and 1e-4 for second-order derivatives. Use deriv init min() to specify the minimum
values as a row vector with two columns; the value in the first column is used for first-order derivatives,
and the value in the second column is used for second-order derivatives. If this function is used,
deriv init usemin() is automatically set to "on".

Complex step method

Let us expand f (p + id) at p using the Taylor series on a complex plane and get the following,

f (p + id) = f (p) + idf ′(p)− d2 f ′′(p)

2!
− id3 f ′′′(p)

3!
+ · · ·

where d is the delta used for computing numerical derivatives and i is the imaginary unit, i2 = −1.

Considering only the imaginary part of the above equation on both sides, we get the following:

Im(f (p + id)) = df ′(p)− d3 f ′′′(p)

3!
+ · · ·

Therefore,

f ′(p) =
Im(f (p + id))

d
+ d2 f ′′′(p)

3!
+ · · ·

deriv() uses the following formula for computing the numerical derivative of f () at p:

f ′(p) =
Im(f (p + id))

d

Please note that there is no subtraction in the formula and hence no error caused by cancellation. In
this method, d can take a fairly small number.

The complex step method is a very accurate method, yet it works only with first-order derivatives with
evaluator type "d" or "t". In addition, because the complex step method uses complex extention of
the evaluator, the evaluator itself has to have and allow complex extension. See Martins, Sturdza, and
Alonso (2003) or Squire and Trapp (1998) for more details about the complex step method.

Type d evaluators

You must write an evaluator function to calculate f () before you can use the derivative functions. The
example we showed above was of what is called a type d evaluator. Let’s stay with that.

The evaluator function we wrote was

void myeval(x, y)
{

y = exp(-x^2 + x - 3)
}

All type d evaluators open the same way,

void evaluator(x, y)

although what you name the arguments is up to you. We named the arguments the way that
mathematicians think, although we could just as well have named them the way that statisticians
think:

void evaluator(b, s)

To avoid favoritism, we will write them as

void evaluator(p, v)

That is, we will think in terms of computing the derivative of v = f (p) with respect to the elements
of p.

Here is the full definition of a type d evaluator:

void evaluator(real rowvector p, v)

where v is the value to be returned:

v: real scalar

evaluator() is to fill in v given the values in p.

evaluator() may return v = . if f () cannot be evaluated at p.

Example of a type d evaluator

We wish to compute the gradient of the following function at p1 = 1 and p2 = 2:

v = exp(−p21 − p22 − p1p2 + p1 − p2 − 3)

Our numerical solution to the problem is
: void eval_d(p, v)
> {
> v = exp(-p[1]^2 - p[2]^2 - p[1]*p[2] + p[1] - p[2] - 3)
> }

: D = deriv_init()

: deriv_init_evaluator(D, &eval_d())

: deriv_init_params(D, (1,2))

: grad = deriv(D, 1)

: grad
1 2

1 -.0000501051 -.0001002102

: (-2*1 - 2 + 1)*exp(-1^2 - 2^2 - 1*2 + 1 - 2 - 3)
-.0000501051

: (-2*2 - 1 - 1)*exp(-1^2 - 2^2 - 1*2 + 1 - 2 - 3)
-.0001002102

For this problem, the elements of the gradient function are given by the following formulas, and we
see that deriv() computed values that are in agreement with the analytical results (to the number
of significant digits being displayed).

dv
dp1

= (−2p1 − p2 + 1) exp(−p21 − p22 − p1p2 + p1 − p2 − 3)

dv
dp2

= (−2p2 − p1 − 1) exp(−p21 − p22 − p1p2 + p1 − p2 − 3)

Type v evaluators

In some statistical applications, you will find type v evaluators more convenient to code than type d
evaluators.

In statistical applications, one tends to think of a dataset of values arranged in matrix X, the rows
of which are observations. The function h(p,X[i,.]) can be calculated for each row separately, and
it is the sum of those resulting values that forms the function f (p) from which we would like to
compute derivatives.

Type v evaluators are for such cases.

In a type d evaluator, you return scalar v = f (p).

In a type v evaluator, you return a column vector, v, such that colsum(v) = f (p).

The code outline for type v evaluators is the same as those for d evaluators. All that differs is that
v, which is a real scalar in the d case, is now a real colvector in the v case.

User-defined arguments

The type v evaluators arise in statistical applications and, in such applications, there are data; that is,
just knowing p is not sufficient to calculate v, g, and H. Actually, that same problem can also arise
when coding type d evaluators.

You can pass extra arguments to evaluators. The first line of all evaluators, regardless of type, is

void evaluator(p, v)

If you code

deriv_init_argument(D, 1, X)

the first line becomes

void evaluator(p, X, v)

If you code

deriv_init_argument(D, 1, X)
deriv_init_argument(D, 2, Y)

the first line becomes

void evaluator(p, X, Y, v)

and so on, up to nine extra arguments. That is, you can specify extra arguments to be passed to your
function.

Example of a type v evaluator

You have the following data:

: x
1

1 .35
2 .29
3 .3
4 .3
5 .65
6 .56
7 .37
8 .16
9 .26

10 .19

You believe that the data are the result of a beta distribution process with fixed parameters alpha
and beta, and you wish to compute the gradient vector and Hessian matrix associated with the log
likelihood at some values of those parameters alpha and beta (a and b in what follows). The formula
for the density of the beta distribution is

density(x) =
Γ(a + b)

Γ(a)Γ(b)
xa−1 (1− x)b−1

In our type v solution to this problem, we compute the gradient and Hessian at a = 0.5 and b = 2.

: void lnbetaden_v(p, x, lnf)
> {
> a = p[1]
> b = p[2]
> lnf = lngamma(a+b) :- lngamma(a) :- lngamma(b) :+
> (a-1)*log(x) :+ (b-1)*log(1:-x)
> }

: D = deriv_init()

: deriv_init_evaluator(D, &lnbetaden_v())

: deriv_init_evaluatortype(D, "v")

: deriv_init_params(D, (0.5, 2))

: deriv_init_argument(D, 1, x) ← important
: deriv(D, 2)
[symmetric]

1 2

1 -116.4988089
2 8.724410052 -1.715062542

: deriv_result_gradient(D)
1 2

1 15.12578465 -1.701917722

Note the following:

1. Rather than calling the returned value v, we called it lnf. You can name the arguments as
you please.

2. We arranged for an extra argument to be passed by coding deriv init argument(D, 1,
x). The extra argument is the vector x, which we listed previously for you. In our function,
we received the argument as x, but we could have used a different name just as we used
lnf rather than v.

3. We set the evaluator type to "v".

Type t evaluators

Type t evaluators are for when you need to compute the Jacobian matrix from a vector-valued
function.

Type t evaluators are different from type v evaluators in that the resulting vector of values should
not be summed. One example is when the function f () performs a nonlinear transformation from the
domain of p to the domain of v.

Example of a type t evaluator

Let’s compute the Jacobian matrix for the following transformation:

v1 = p1 + p2

v2 = p1 − p2

Here is our numerical solution, evaluating the Jacobian at p = (0, 0):

: void eval_t1(p, v)
> {
> v = J(1,2,.)
> v[1] = p[1] + p[2]
> v[2] = p[1] - p[2]
> }

: D = deriv_init()

: deriv_init_evaluator(D, &eval_t1())

: deriv_init_evaluatortype(D, "t")

: deriv_init_params(D, (0,0))

: deriv(D, 1)
[symmetric]

1 2

1 1
2 1 -1

Now let’s compute the Jacobian matrix for a less trivial transformation:

v1 = p21

v2 = p1 p2

Here is our numerical solution, evaluating the Jacobian at p = (1, 2):

: void eval_t2(p, v)
> {
> v = J(1,2,.)
> v[1] = p[1]^2
> v[2] = p[1] * p[2]
> }

: D = deriv_init()

: deriv_init_evaluator(D, &eval_t2())

: deriv_init_evaluatortype(D, "t")

: deriv_init_params(D, (1,2))

: deriv(D, 1)
1 2

1 1.999999998 0
2 2 1

Example of using step-size lower bounds

Sometimes, the numerical differentiation method gives unstable results when the computed step size,
delta, is too small. In this case, we can set lower bounds for step sizes. Let’s consider the derivative
of the normal density function with respect to the mean and standard deviation at m = 0, s = 0.1,
and x = 0.5.

: void f(real vector params, real scalar x, real scalar v)
> {
> m = params[1]
> s = params[2]
> v = normalden(x, m, s)
> }

:
: // exact derivatives
: m = 0

: s = 0.1

: x = 0.5

The exact solution for the Hessian matrix is

: Hessian
[symmetric]

1 2

1 .0356812684
2 .1635391466 .7463331964

Without setting the step-size lower bounds, we see the computed result with adaptive step length
search is

: D = deriv_init()

: deriv_init_evaluator(D, &f())

: deriv_init_evaluatortype(D, "d")

:
: deriv_init_params(D, params)

: deriv_init_argument(D, 1, x)

:
: d2_nousemin = deriv(D, 2)

: d2_nousemin
[symmetric]

1 2

1 .0788226616
2 -.026272616 1.84061274

After setting the step-size lower bounds, we see the result is much closer to the exact solution:

: deriv_init_usemin(D, "on")

:
: d2_usemin = deriv(D, 2)

: d2_usemin
[symmetric]

1 2

1 .0356813276
2 .1635502121 .7463480108

The relative difference between this result and the exact solution is

: mreldif(d2_usemin, Hessian)
9.51017e-06

Example of complex step method

We can use the complex step method to compute first-order derivatives with complex extension. For
example, let’s consider the derivative of the following function:

v = p2.5

Here we set up the problem and choose the complex step method in deriv init technique():

: void eval(p, v) {
> v = p ^ 2.5
> }

:
: D = deriv_init()

: deriv_init_evaluator(D, &eval())

: deriv_init_technique(D, "complex")

Now we evaluate the gradient at p = 0.5:

: deriv_init_params(D, 0.5)

: grad = deriv(D, 1)

: grad
.8838834765

Functions

deriv init( )

transmorphic deriv init()

deriv init() is used to begin a derivative problem. Store the returned result in a variable name
of your choosing; we have used D in this documentation. You pass D as the first argument to the
other deriv*() functions.

deriv init() sets all deriv init *() values to their defaults. You may use the query form of
deriv init *() to determine an individual default, or you can use deriv query() to see them
all.

The query form of deriv init *() can be used before or after calling deriv().

deriv init evaluator( ) and deriv init evaluatortype( )

void deriv init evaluator(D, pointer(function) scalar fptr)

void deriv init evaluatortype(D, evaluatortype)

pointer(function) scalar deriv init evaluator(D)

string scalar deriv init evaluatortype(D)

deriv init evaluator(D, fptr) specifies the function to be called to evaluate f (p). Use of this
function is required. If your function is named myfcn(), you code deriv init evaluator(D,
&myfcn()).

deriv init evaluatortype(D, evaluatortype) specifies the capabilities of the function that has
been set using deriv init evaluator(). Alternatives for evaluatortype are "d", "v", and "t".
The default is "d" if you do not invoke this function.

deriv init evaluator(D) returns a pointer to the function that has been set.

deriv init evaluatortype(D) returns the evaluator type currently set.

deriv init argument( ) and deriv init narguments( )

void deriv init argument(D, real scalar k, X)

void deriv init narguments(D, real scalar K)

pointer scalar deriv init argument(D, real scalar k)

real scalar deriv init narguments(D)

deriv init argument(D, k, X) sets the kth extra argument of the evaluator function to be X. X
can be anything, including a view matrix or even a pointer to a function. No copy of X is made; it
is a pointer to X that is stored, so any changes you make to X between setting it and X being used
will be reflected in what is passed to the evaluator function.

deriv init narguments(D, K) sets the number of extra arguments to be passed to the evaluator
function. This function is useless and included only for completeness. The number of extra arguments
is automatically set when you use deriv init argument().

deriv init argument(D, k) returns a pointer to the object that was previously set.

deriv init narguments(D) returns the number of extra arguments that were passed to the
evaluator function.

deriv init weights( )

void deriv init weights(D, real colvector weights)

pointer scalar deriv init weights(D)

deriv init weights(D, weights) sets the weights used with type v evaluators to produce the
function value. By default, deriv() with a type v evaluator uses colsum(v) to compute the function

value. With weights, deriv() uses cross(weights, v). weights must be row conformable with the
column vector returned by the evaluator.

deriv init weights(D) returns a pointer to the weight vector that was previously set.

deriv init params( )

void deriv init params(D, real rowvector params)

real rowvector deriv init params(D)

deriv init params(D, params) sets the parameter values at which the derivatives will be
computed. Use of this function is required.

deriv init params(D) returns the parameter values at which the derivatives were computed.

Advanced init functions

The rest of the deriv init *() functions provide finer control of the numerical derivative taker.

deriv init h( ), . . . scale( ), . . . bounds( ), . . . search( )

void deriv init h(D, real rowvector h)

void deriv init scale(D, real rowvector s)

void deriv init bounds(D, real rowvector minmax)

void deriv init search(D, search)

real rowvector deriv init h(D)

real rowvector deriv init scale(D)

real rowvector deriv init bounds(D)

string scalar deriv init search(D)

deriv init h(D, h) sets the h values used to compute numerical derivatives.

deriv init scale(D, s) sets the starting scale values used to compute numerical derivatives.

deriv init bounds(D, minmax) sets the minimum and maximum values used to search for
optimal scale values. The default is minmax = (1e-8, 1e-7).

deriv init search(D, "interpolate") causes deriv() to use linear and quadratic interpola-
tion to search for an optimal delta for computing the numerical derivatives. This is the default search
method.

deriv init search(D, "bracket") causes deriv() to use a bracketed quadratic formula to
search for an optimal delta for computing the numerical derivatives.

deriv init search(D, "off") prevents deriv() from searching for an optimal delta.

deriv init h(D) returns the user-specified h values.

deriv init scale(D) returns the user-specified starting scale values.



16 deriv( ) — Numerical derivatives

deriv init bounds(D) returns the user-specified search bounds.

deriv init search(D) returns the currently set search method.

These functions only work with the numerical differentiation method.

deriv init usemin( ) and deriv init min( )

void deriv init usemin(D, { "off" | "on" })
void deriv init min(D, real rowvector min)

string scalar deriv init usemin(D)

real rowvector deriv init min(D)

deriv init usemin(D, { "off" | "on" }) specifies whether to use the minimum values of the
step size, delta, for computing numerical derivatives. The default is "off", meaning minimum values
will not be used. If you specify "on", minimum values will be used; when minimum values are
not specified by deriv init min(), default minimum values are 1e-6 and 1e-4 for first- and
second-order derivatives, respectively.

deriv init min(D, min) sets the minimum values of the delta used for computing numerical
derivatives. min is a 1 × 2 real row vector; the first column specifies the minimum for first-order
derivatives, and the second column specifies the minimum for second-order derivatives. If there is a
missing value in the row vector, the default values of 1e-6 for first-order derivatives and 1e-4 for
second-order derivatives are used. If this function is used, deriv init usemin() is automatically
set to "on".

deriv init usemin(D) returns the current value "off" or "on", indicating whether to use
minimum values of the step size.

deriv init min(D) returns the current value min.

These functions only work with the numerical differentiation method.

deriv init verbose( )

void deriv init verbose(D, verbose)

string scalar deriv init verbose(D)

deriv init verbose(D, verbose) sets whether error messages that arise during the execution of
deriv() or deriv() are to be displayed. Setting verbose to "on" means that they are displayed;
"off" means that they are not displayed. The default is "on". Setting verbose to "off" is of interest
only to users of deriv().

deriv init verbose(D) returns the current value of verbose.

deriv init technique( )

(varies) deriv init technique(D, technique)

deriv init technique(technique) specifies the technique used to compute the numerical deriva-
tives.

The numerical differentiation method ("numdiff") is the default setting for technique.

The complex step method ("complex") requires that the evaluator be extended to complex numbers.
This method can be used with first-order derivatives when the evaluator type is "d" or "t".

deriv init technique(D) returns the current technique.

deriv( )

(varies) deriv(D, todo)

deriv(D, todo) invokes the derivative process. If something goes wrong, deriv() aborts with
error.

deriv(D, 0) returns the function value without computing derivatives.

deriv(D, 1) returns the gradient vector; the Hessian matrix is not computed.

deriv(D, 2) returns the Hessian matrix; the gradient vector is also computed.

Before you can invoke deriv(), you must have defined your evaluator function, evaluator(), and
you must have set the parameter values at which deriv() is to compute derivatives:

D = deriv init()

deriv init evaluator(D, &evaluator())

deriv init params(D, (. . .))

The above assumes that your evaluator function is type d. If your evaluator function type is v (that
is, it returns a column vector of values instead of a scalar value), you will also have coded

deriv init evaluatortype(D, "v")

and you may have coded other deriv init *() functions as well.

Once deriv() completes, you may use the deriv result *() functions. You may also continue
to use the deriv init *() functions to access initial settings, and you may use them to change
settings and recompute derivatives (that is, invoke deriv() again) if you wish.

deriv( )

real scalar deriv(D, todo)

deriv(D) performs the same actions as deriv(D) except that, rather than returning the requested
derivatives, deriv() returns a real scalar and, rather than aborting if numerical issues arise,
deriv() returns a nonzero value. deriv() returns 0 if all went well. The returned value is called

an error code.

deriv() returns the requested result. It can work that way because the numerical derivative calculation
must have gone well. Had it not, deriv() would have aborted execution.

deriv() returns an error code. If it is 0, the numerical derivative calculation went well, and you
can obtain the gradient vector by using deriv result gradient(). If things did not go well, you
can use the error code to diagnose what went wrong and take the appropriate action.

Thus deriv(D) is an alternative to deriv(D). Both functions do the same thing. The difference
is what happens when there are numerical difficulties.

deriv() and deriv() work around most numerical difficulties. For instance, the evaluator function
you write is allowed to return v equal to missing if it cannot calculate the f ( ) at p + d. If that happens
while computing the derivative, deriv() and deriv() will search for a better d for taking the
derivative. deriv(), however, cannot tolerate that happening at p (the parameter values you set
using deriv init params()) because the function value must exist at the point when you want
deriv() to compute the numerical derivative. deriv() issues an error message and aborts, meaning
that execution is stopped. There can be advantages in that. The calling program need not include
complicated code for such instances, figuring that stopping is good enough because a human will
know to address the problem.

deriv(), however, does not stop execution. Rather than aborting, deriv() returns a nonzero
value to the caller, identifying what went wrong. The only exception is that deriv() will return
a zero value to the caller even when the evaluator function returns v equal to missing at p, allowing
programmers to handle this special case without having to turn deriv init verbose() off.

Programmers implementing advanced systems will want to use deriv() instead of deriv().
Everybody else should use deriv().

Programmers using deriv() will also be interested in the functions deriv init verbose(),
deriv result errorcode(), deriv result errortext(), and
deriv result returncode().

The error codes returned by deriv() are listed below, under the heading deriv result errorcode( ),
. . . errortext( ), and . . . returncode( ).

deriv result value( )

real scalar deriv result value(D)

deriv result value(D) returns the value of f () evaluated at p.

deriv result values( ) and deriv result values( )

real matrix deriv result values(D)

void deriv result values(D, v)

deriv result values(D) returns the vector values returned by the evaluator. For type v evaluators,
this is the column vector that sums to the value of f () evaluated at p. For type t evaluators, this is
the rowvector returned by the evaluator.

deriv result values(D, v) uses swap() (see [M-5] swap( )) to interchange v with the vector
values stored in D. This destroys the vector values stored in D.

These functions should be called only with type v evaluators.

deriv result gradient( ) and deriv result gradient( )

real rowvector deriv result gradient(D)

void deriv result gradient(D, g)

deriv result gradient(D) returns the gradient vector evaluated at p.

deriv result gradient(D, g) uses swap() (see [M-5] swap( )) to interchange g with the
gradient vector stored in D. This destroys the gradient vector stored in D.

deriv result scores( ) and deriv result scores( )

real matrix deriv result scores(D)

void deriv result scores(D, S)

deriv result scores(D) returns the matrix of the scores evaluated at p. The matrix of scores
can be summed over the columns to produce the gradient vector.

deriv result scores(D, S) uses swap() (see [M-5] swap( )) to interchange S with the scores
matrix stored in D. This destroys the scores matrix stored in D.

These functions should be called only with type v evaluators.

deriv result Jacobian( ) and deriv result Jacobian( )

real matrix deriv result Jacobian(D)

void deriv result Jacobian(D, J)

deriv result Jacobian(D) returns the Jacobian matrix evaluated at p.

deriv result Jacobian(D, J) uses swap() (see [M-5] swap( )) to interchange J with the
Jacobian matrix stored in D. This destroys the Jacobian matrix stored in D.

These functions should be called only with type t evaluators.

https://www.stata.com/manuals/m-5swap.pdf#m-5swap()
https://www.stata.com/manuals/m-5swap.pdf#m-5swap()
https://www.stata.com/manuals/m-5swap.pdf#m-5swap()
https://www.stata.com/manuals/m-5swap.pdf#m-5swap()

deriv result Hessian( ) and deriv result Hessian( )

real matrix deriv result Hessian(D)

void deriv result Hessian(D, H)

deriv result Hessian(D) returns the Hessian matrix evaluated at p.

deriv result Hessian(D, H) uses swap() (see [M-5] swap( )) to interchange H with the
Hessian matrix stored in D. This destroys the Hessian matrix stored in D.

These functions should not be called with type t evaluators.

deriv result h( ), . . . scale( ), and . . . delta( )

real rowvector deriv result h(D)

real rowvector deriv result scale(D)

real rowvector deriv result delta(D)

deriv result h(D) returns the vector of h values that was used to compute the numerical
derivatives.

deriv result scale(D) returns the vector of scale values that was used to compute the numerical
derivatives.

deriv result delta(D) returns the vector of delta values used to compute the numerical deriva-
tives.

deriv result errorcode( ), . . . errortext( ), and . . . returncode( )

real scalar deriv result errorcode(D)

string scalar deriv result errortext(D)

real scalar deriv result returncode(D)

These functions are for use after deriv().

deriv result errorcode(D) returns the same error code as deriv(). The value will be zero
if there were no errors. The error codes are listed in the table directly below.

deriv result errortext(D) returns a string containing the error message corresponding to the
error code. If the error code is zero, the string will be "".

deriv result returncode(D) returns the Stata return code corresponding to the error code. The
mapping is listed in the table directly below.

In advanced code, these functions might be used as
(void) _deriv(D, todo)
. . . if (ec = deriv_result_code(D)) {

errprintf("{p}\n")
errprintf("%s\n", deriv_result_errortext(D))
errprintf("{p_end}\n")
exit(deriv_result_returncode(D))
/*NOTREACHED*/

}

https://www.stata.com/manuals/m-5swap.pdf#m-5swap()

The error codes and their corresponding Stata return codes are
Error Return
code code Error text

1 198 invalid todo argument

2 111 evaluator function required

3 459 parameter values required

4 459 parameter values not feasible

5 459 could not calculate numerical derivatives—discontinuous region with missing
values encountered

6 459 could not calculate numerical derivatives—flat or discontinuous region
encountered

16 111 function() not found

17 459 Hessian calculations not allowed with type t evaluators

Note: Error 4 can occur only when evaluating f () at the parameter values.
This error occurs only with deriv().

deriv query( )

void deriv query(D)

deriv query(D) displays a report on the current deriv init *() values and some of the
deriv result *() values. deriv query(D) may be used before or after deriv(), and it is
useful when using deriv() interactively or when debugging a program that calls deriv() or
deriv().

Conformability

All functions have 1× 1 inputs and have 1× 1 or void outputs, except the following:

deriv init params(D, params):
D: transmorphic

params: 1× np
result: void

deriv init params(D):
D: transmorphic

result: 1× np

deriv init argument(D, k, X):
D: transmorphic
k: 1× 1
X: anything

result: void

deriv init weights(D, params):
D: transmorphic

params: N× 1
result: void

deriv init h(D, h):
D: transmorphic
h: 1× np

result: void

deriv init h(D):
D: transmorphic

result: 1× np

deriv init scale(D, scale):
D: transmorphic

scale: 1× np (type d and v evaluator)
nv× np (type t evaluator)

void: void

deriv init bounds(D, minmax):
D: transmorphic

minmax: 1× 2
result: void

deriv init bounds(D):
D: transmorphic

result: 1× w

deriv init min(D, min):
D: transmorphic

min: 1× 2
result: void

deriv init min(D):
D: transmorphic

result: 1× 2

deriv(D, 0):
D: transmorphic

result: 1× 1
1× nv (type t evaluator)

deriv(D, 1):
D: transmorphic

result: 1× np
nv× np (type t evaluator)

deriv(D, 2):
D: transmorphic

result: np× np

deriv result values(D):
D: transmorphic

result: N × 1
1× nv (type t evaluator)
N× 1 (type v evaluator)

deriv result values(D, v):
D: transmorphic
v: N× 1

1× nv (type t evaluator)
N× 1 (type v evaluator)

result: void

deriv result gradient(D):
D: transmorphic

result: 1× np

deriv result gradient(D, g):
D: transmorphic
g: 1× np

result: void

deriv result scores(D):
D: transmorphic

result: N× np

deriv result scores(D, S):
D: transmorphic
S: N× np

result: void

deriv result Jacobian(D):
D: transmorphic

result: nv× np

deriv result Jacobian(D, J):
D: transmorphic
J: nv× np

result: void

deriv result Hessian(D):
D: transmorphic

result: np× np

deriv result Hessian(D, H):
D: transmorphic
H: np× np

result: void



24 deriv( ) — Numerical derivatives

deriv result h(D):
D: transmorphic

result: 1× np

deriv result scale(D):
D: transmorphic

result: 1× np (type d and v evaluator)
nv× np (type t evaluator)

deriv result delta(D):
D: transmorphic

result: 1× np (type d and v evaluator)
nv× np (type t evaluator)

Diagnostics

All functions abort with error when used incorrectly.

deriv() aborts with error if it runs into numerical difficulties. deriv() does not; it instead returns
a nonzero error code.

Methods and formulas

See sections 1.3.4 and 1.3.5 of Pitblado, Poi, and Gould (2024) for an overview of the methods and
formulas deriv() uses to compute numerical derivatives.

� �
Carl Gustav Jacob Jacobi (1804–1851) was born in Potsdam, Prussia, which is now Germany.
Jacobi demonstrated great potential at a young age, meeting the qualifications to enter university
at the age of 12. He obtained his PhD from the University of Berlin in 1825. In 1829, Jacobi
published a paper making significant contributions to elliptic functions, which were consequently
named after him. He also published a paper on functional determinants, giving rise to the
Jacobian matrix and Jacobian determinant. He made many contributions to number theory,
including proofs of quadratic reciprocity and Fermat’s two-square theorem. In addition to the
functions and equations named after him, Jacobi was honored with a crater in his name. In 1833,
the Royal Society elected Jacobi to fellowship, as did the Royal Society of Edinburgh in 1845.

Jacobi spent 15 years teaching at Königsberg University, where he also supervised the doctoral
work of Ludwig Otto Hesse. Because of health complications, he spent some time in Italy and
later relocated to Berlin, where he gave some lectures. However, after making some unfavorable
political statements, he was denied the opportunity to work at the University of Berlin. This
decision was later retracted, and he left his position at the University of Vienna to lecture at the
University of Berlin. Jacobi died of smallpox in 1851.� �

